

# ODE and DAE Solvers

---

## Advanced Topics from Scientific Computing

TU Berlin Winter 2024/25

### Notebook 09

 Jürgen Fuhrmann

GRBackend()

```
1 begin
2   using LinearAlgebra: norm, I, Diagonal
3   using ForwardDiff: ForwardDiff
4   using DiffResults: DiffResults
5   using NLSolve: nlsolve, converged
6   using RecursiveArrayTools: DiffEqArray
7   using Plots: Plots, plot, plot!, contourf
8   using Plots: hline!, vline!, grid
9   Plots.gr()
10 end
```

#### Literature:

- E. Hairer, S.P. Nørsett, G. Wanner: Solving Ordinary Differential Equations I. Nonstiff Problems
- E. Hairer, G. Wanner: Solving Ordinary Differential Equations II. Stiff and Differential Algebraic Problems
- P. Kunkel, V. Mehrmann: Differential-Algebraic Equations. Analysis and Numerical Solution

#### Further unpublisued sources used:

- TU Chemnitz lecture notes by A. Naumann
- HU Berlin course "Numerik gewöhnlicher Differentialgleichungen I" (implementation of Explicit Runge-Kutta)

While this course focuses on the solution of partial differential equations which describe coupled processes in space and time, we first talk about the solution of ordinary differential equations, providing methods for discretization in time.

Look for  $u : [0, T] \rightarrow \mathbb{R}^m$  be a time dependent, differentiable function such that it fulfills the initial value problem for the ordinary differential equation (ODE) system:

$$\begin{cases} \dot{u} & = f(u, p, t) \\ u|_{t=0} & = u_0 \end{cases}$$

with  $f : \mathbb{R}^m \times \mathbb{R}^k \times [0, T] \rightarrow \mathbb{R}^m$ .

Here,  $p$  is a  $k$ -vector of parameters.

The system is called *autonomous* if  $f$  does not depend on  $t$ .

Subdivide  $[0, T]$  into  $N - 1$  intervals of time step size  $\tau = \frac{T}{N-1}$ . Let  $t_n = (n - 1)\tau$  and let  $u_n = u(t_n)$ .

## Explicit Euler method

The simplest way of discretizing in time is to equate the finite difference in time to the right hand side value calculated in the last timestep.

$$\frac{u_{n+1} - u_n}{\tau} = f(u_n, p, t_n)$$

We implement this *explicit Euler method* such that we describe the right hand side  $du=f(u, p, t)$  via a mutating function  $f!(du, u, p, t)$ . It returns a vector of solution vectors and the chosen time values.

```

1 function explicit_euler(
2     f!::F, # ODE right hand side
3     p; # Parameters
4     u0 = [0.0], # vector of initial values
5     dt = 0.1, # time step size
6     tspan = [0.0, 1.0],
7 ) where {F}
8     u = [u0]
9     t = Float64[tspan[1]]
10    du = zeros(length(u0))
11    while t[end] < tspan[end]
12        uold = u[end]
13        told = t[end]
14        f!(du, uold, p, told)
15        unew = uold + dt * du
16        tnew = told + dt
17        push!(u, unew)
18        push!(t, tnew)
19    end
20    return u, t
21 end;
```

## Test problem

Let us first look at the equation

$$\dot{u} = \lambda u$$

```
1 f_test!(du, u, λ, t) = du[1] = λ * u[1];
```

The exact solution is given by  $u = u_0 e^{\lambda t}$ .

```
1 exact_test(u0, p, t) = u0 * exp(p * t);
```

For  $\lambda = \frac{\log \frac{1}{2}}{t_{\frac{1}{2}}}$ , this equation e.g. describes radioactive decay with half value time  $t_{\frac{1}{2}}$


```
p_test = -5
```

```
1 p_test = -5
```

```
(
  1:  [[1.0], [-0.655656], [0.429884], [-0.281856], [0.1848], [-0.121165], [0.0794428],
  2:  [0.0, 0.331131, 0.662262, 0.993393, 1.32452, 1.65566, 1.98679, 2.31792, 2.64904,
)
```

```
1 u_test, t_test = explicit_euler(
2   f_test!, p_test; u0 = [1.0], dt = τ,
3   tspan = (0, 10)
4 )
```

$\tau =$   0.3311311214825911



Obviously we may have a problem with the step size, and we may try to find other methods.

## θ- and Runge-Kutta methods

Main theorem of calculus gives

$$\int_{t_n}^{t_{n+1}} \dot{u} = \int_{t_n}^{t_{n+1}} f(u(t), p, t)$$

Approximate the integral using a quadrature rule with points  $c_i$  and weights  $b_i$ :

$$u_{n+1} - u_n = \tau \sum_{i=1}^s b_i f(u_{n,i}, p, t_n + c_i \tau)$$

Using some simple quadratures leads to

$$b = \{1\}, c = \{\theta\}, u_{n,1} = \theta u_{n+1} + (1 - \theta) u_n$$

- $\theta = 0$ : explicit Euler method
- $\theta = 1$ : implicit Euler method
- $\theta = \frac{1}{2}$ : implicit midpoint method

$$\frac{u_{n+1} - u_n}{\tau} = f(\theta u_{n+1} + (1 - \theta) u_n, p, t_n + \theta \tau)$$

Unless  $\theta=0$  (explicit Euler method) we need to solve an equation or a system of equations in each timestep. Here we implement this for general  $\theta$ .

## $\theta$ method: implementation

This implementation is slightly more general:

- if `linear=true`, instead of Newton's method to solve the implicit equations it performs just one Newton step (assuming the time step size is small enough). This results in a *linear implicit* method (linear implicit Euler or linear implicit midpoint)
- it implements the solution of a slightly more general problem: for a matrix  $M$  (unit matrix by default), solve

$$M\dot{u} = f(u, p, t).$$

```

1 function θmethod(
2     f!::F, # right hand side function
3     p; # Parameters
4     u0 = [0.0], # Initial value
5     dt = 0.1, # time step size
6     tspan = (0, 1), # time interval
7     θ = 0, # θ parameter
8     linear = false, # switch between linear method and newton solver
9     M = I, # Mass matrix
10    tol = 1.0e-10,
11 ) where {F}
12 # Start arrays for the result
13 u = [u0]
14 t = Float64[tspan[1]]
15 uold = u[end]
16 told = t[end]
17
18 # Each timestep involves the solution of F(uneu)=0
19 # Here, uneu is the current iteration and F is the result
20 function fstep!(F, uneu)
21     f!(F, θ * uneu + (1 - θ) * uold, p, told + θ * dt)
22     return F .- M * (uneu .- uold) ./ dt .- F
23 end
24
25 if linear
26     # Handle differentiation results
27     diffresult = DiffResults.JacobianResult(u0)
28     y = zero(u0)
29     cfg = ForwardDiff.JacobianConfig(fstep!, y, u0)
30 end
31
32 while t[end] < tspan[2]
33     # Loop over the timesteps
34     told = t[end]
35     uold = u[end]
36     if linear
37         # Perform one Newton step
38         ForwardDiff.jacobian!(diffresult, fstep!, y, uold, cfg)
39         uneu = uold -
40             DiffResults.jacobian(diffresult) \
41             DiffResults.value(diffresult)
42     else
43         # Solve time step problem and check for convergence
44         result = nlsolve(fstep!, uold; autodiff = :forward, xtol = tol)
45         !converged(result) && throw("convergence error")
46         uneu = result.zero
47     end
48     # Extend solution arrays
49     push!(u, uneu)
50     push!(t, told + dt)
51 end
52 # Return the solution as DiffEqArray
53 # from RecursiveArrayTools.jl (easy to plot)
54 return DiffEqArray(u, t)
55 end;

```

So let us use this for our test example:


uθ\_test =

	timestamp	value1
1	0.0	1.0
2	0.143845	0.471007
3	0.28769	0.221848
4	0.431535	0.104492
5	0.57538	0.0492165
6	0.719225	0.0231813
7	0.86307	0.0109186
8	1.00691	0.00514273
9	1.15076	0.00242227
10	1.2946	0.00114091

more

```
1 uθ_test = θmethod(f_test!, p_test; u0 = [1.0], dt = τ2, tspan = (0, 10), θ = θx)
```

θ=  0.5 τ=  0.14384498882876628



## Runge-Kutta methods

A generalization of the the choice of internal stages  $u_{n,i}$  in the quadrature based method starting with

$$\int_{t_n}^{t_{n+1}} \dot{u} = \int_{t_n}^{t_{n+1}} f(u(t), p, t)$$

$$u_{n+1} - u_n = \tau \sum_{i=1}^s b_i f(u_{n,i}, p, t_n + c_i \tau)$$

chooses another quadrature rule with the points  $c_1 \dots c_s$  and weights  $a_{ij}$ :

$$u_{n,i} - u_n = \tau \sum_{j=1}^s a_{ij} f(u_{n,j}, p, t_n + c_j \tau)$$

This Ansatz gives s-stage Runge-Kutta methods.

If  $A$  is strictly lower triangular, the method is explicit, i.e. it does only involve the evaluation of  $f$ , no solution of a system of equations involving  $f$ .

Otherwise, it is implicit, and it involves the solution of one or more, possibly coupled systems of equations of dimension  $m \times s$ . As a consequence, implicit Runge-Kutta methods are hard to use for large systems.

Runge-Kutta methods thus are described by the points  $c = \{c_1 \dots c_s\}$ , the weights  $b = \{b_1 \dots b_s\}$  and the  $s \times s$  matrix  $A = (a_{ij})$ . Usually these are arranged into *Butcher tableaux*:

c	A
	b

These tableaux can be conveniently managed using the `RungeKutta.jl` package.

```

1 begin
2     using RungeKutta: RungeKutta, Tableau
3     using RungeKutta: TableauExplicitEuler, TableauImplicitEuler
4     using RungeKutta: TableauImplicitMidpoint, TableauRK4
5     using RungeKutta: TableauRadauIIA
6 end

```

The  $\theta$  method is a Runge-Kutta method:

Runge-Kutta Tableau ExplicitEuler with 1 stage and order 1:

0//1	0//1
	1//1

```

1 TableauExplicitEuler(Rational)

```

Runge-Kutta Tableau ImplicitEuler with 1 stage and order 1:

1//1	1//1
1//1	1//1

1 [TableauImplicitEuler\(Rational\)](#)

Runge-Kutta Tableau ImplicitMidpoint with 1 stage and order 2:

1//2	1//2
1//1	1//1

1 [TableauImplicitMidpoint\(Rational\)](#)

The classical Runge-Kutta method is "RK4":

Runge-Kutta Tableau RK416 with 4 stages and order 4:

0//1	0//1	0//1	0//1	0//1
1//2	1//2	0//1	0//1	0//1
1//2	0//1	1//2	0//1	0//1
1//1	0//1	0//1	1//1	0//1
	1//6	1//3	1//3	1//6

1 [TableauRK4\(Rational\)](#)

With RungeKutta.jl, one can e.g. define the Dormand-Prince tableau:



TableauDP (generic function with 4 methods)

```

1 begin
2   function TableauDP(T = Float64, order = 4)
3     A = T[
4       0 0 0 0 0 0 0
5       1 // 5 0 0 0 0 0 0
6       3 // 40 9 // 40 0 0 0 0 0
7       44 // 45 -56 // 15 32 // 9 0 0 0 0
8       19372 // 6561 -25360 // 2187 64448 // 6561 -212 // 729 0 0 0
9       9017 // 3168 -355 // 33 46732 // 5247 49 // 176 -5103 // 18656 0 0
10      35 // 384 0 500 // 1113 125 // 192 -2187 // 6784 11 // 84 0
11    ]
12    c = T[0; 1 // 5; 3 // 10; 4 // 5; 8 // 9; 1; 1]
13    b4 = T[
14      5179 // 57600
15      0
16      7571 // 16695
17      393 // 640
18      -92097 // 339200
19      187 // 2100
20      1 // 40
21    ]
22    b5 = T[
23      35 // 384; 0; 500 // 1113; 125 // 192; -2187 // 6784; 11 // 84;
24      0
25    ]
26
27    return if order == 4
28      RungeKutta.Tableau(:DoPri4, order, A, b4, c)
29    elseif order == 5
30      RungeKutta.Tableau(:DoPri5, order, A, b5, c)
31    else
32      throw(error("DormandPrince can have orders 4 or 5"))
33    end
34  end
35  TableauDP(o::Number) = TableauDP(Float64, o)
36 end

```

Runge-Kutta Tableau DoPri5 with 7 stages and order 5:

0//1	0//1	0//1	0//1	0//1	0//1
1//5	1//5	0//1	0//1	0//1	0//1
3//10	3//40	9//40	0//1	0//1	0//1
4//5	44//45	-56//15	32//9	0//1	0//1
8//9	19372//6561	-25360//2187	64448//6561	-212//729	0//1
1//1	9017//3168	-355//33	46732//5247	49//176	-5103//18656
1//1	35//384	0//1	500//1113	125//192	-2187//6784
	35//384	0//1	500//1113	125//192	-2187//6784

2 columns omitted

```
1 TableauDP(Rational, 5)
```

The Radau IIA method is an implicit Runge-Kutta method:

Runge-Kutta Tableau RadauIIA with 3 stages and order 5:

0.155051	0.196815	-0.0655354	0.023771
0.644949	0.394424	0.292073	-0.0415488
1.0	0.376403	0.512486	0.111111
	0.376403	0.512486	0.111111

1 [TableauRadauIIA\(3\)](#)

Implementation of the explicit Runge-Kutta scheme:

explicit\_rk (generic function with 1 method)

```

1 function explicit_rk(
2     f!::F, # ODE right hand side
3     p; # Parameters
4     tableau = TableauRK4\(\),
5     u0 = [0.0], # vector of initial values
6     dt = 0.1, # time step size
7     tspan = (0, 1),
8 ) where {F}
9     u = [u0]
10    t = Float64[tspan[1]]
11    a = tableau.a
12    b = tableau.b
13    c = tableau.c
14    m = length(u0)
15    s = length(b)
16    du = zeros(m)
17    z = zeros(m, s)
18    while t[end] < tspan[end]
19        z .= 0
20        for j in 1:s
21            du .= 0
22            for k in 1:(j - 1)
23                @views du .+= z[:, k] * a[j, k]
24            end
25            @views f!(z[:, j], u[end] .+ dt * du, p, t[end] + c[j] * dt)
26        end
27        du .= 0
28        for j in 1:s
29            @views du .+= b[j] * z[:, j]
30        end
31        u_new = u[end] + dt * du
32        t_new = t[end] + dt
33        push!(u, u_new)
34        push!(t, t_new)
35    end
36    return DiffEqArray(u, t)
37 end

```

1 tableau = [TableauRK4\(\)](#);

urk\_test =

---


	timestamp	value1
<b>1</b>	0.0	1.0
<b>2</b>	0.475081	0.538578
<b>3</b>	0.950162	0.290066
<b>4</b>	1.42524	0.156223
<b>5</b>	1.90032	0.0841384
<b>6</b>	2.37541	0.0453151
<b>7</b>	2.85049	0.0244057
<b>8</b>	3.32557	0.0131444
<b>9</b>	3.80065	0.00707927
<b>10</b>	4.27573	0.00381274

more

---

```
1 urk_test = explicit_rk(  
2   f_test!, p_test; u0 = [1.0], tableau, dt = trk,  
3   tspan = (0, 10)  
4 )
```

$\tau =$   0.47508101621027965



```

1 let
2   p = plot(
3     framestyle = :box,
4     gridlinewidth = 2,
5     size = (600, 300),
6     xlabel = "t",
7     ylabel = "solution", ylimits = (-1.5, 1.5),
8     title="τ=$(round(τrk, sigdigits=4))",
9   )
10  plot!(p, urk_test; linewidth = 2, label = "numerical")
11  T = urk_test.t
12  plot!(p, T, exact_test.(1.0, p_test, T); label = "exact", linewidth = 2)
13 end

```

Once again we see that for a "decent" solution, the time step size should be not too large. We need to discuss two phenomena:

- What about the "wiggles" which destroy the overall quality of solutions  $\Rightarrow$  stability
- How good is the approximation  $\Rightarrow$  approximation order

## Stability

Let us discuss aspects of stability (resp. instability).

Accidentally, the test problem  $\dot{u} = \lambda u$  chosen here is also called *Dahlquist test equation*. It is used to judge the stability of ODE methods.

## A-Stability

**Definition:** An ODE method is called unconditionally A-stable if for  $\lambda \in \mathbb{C}$  with  $\text{Re}\lambda < 0$ , the approximate solution of the Dahlquist test equation fulfills

$$|u_{n+1}| \leq |u_n|.$$

The method is called conditionally A stable if  $|u_{n+1}| \leq |u_n|$  is fulfilled for  $h\lambda \in S \subset \mathbb{C}$ , where  $S$  is called stability region.

In order to investigate A-Stability, it is useful to define the stability function  $R$  such that  $u_{n+1} = R(\tau\lambda)(u_n)$

For the  $\theta$ -methods this gives:

```
1 R_0(z) = 1 + z;
```

```
1 R_05(z) = (1 + z / 2) / (1 - z / 2);
```

```
1 R_1(z) = 1 / (1 - z);
```

For generic Runge-Kutta methods, we have:

```
1 function R_rk(z, tableau = TableauRK4())
2     A = tableau.a
3     b = tableau.b
4     Ib = ones(length(b))
5     return 1 + z * b' * inv(I - z * A) * Ib
6 end;
```

## L-Stability

An A-stable one-step method is called L-stable if for all  $\lambda \in \mathbb{C}$  with  $\text{Re}\lambda < 0$ , the solution of the Dahlquist test problem fulfils

$$\lim_{n \rightarrow \infty} |u_n| = 0$$

independent of the stepsize  $\tau$ .

L-Stability is fulfilled if and only if for all  $\lambda \in \mathbb{C}$  with  $\text{Re}\lambda < 0$ ,

$$\lim_{\tau \rightarrow \infty} R(\tau\lambda) = 0$$

We can visualize the stability region by plotting the range where  $R < 1$  in red:


stabregion (generic function with 1 method)

```

1 function stabregion(
2     R;
3     xrange = [-3, 3],
4     yrange = [-3, 3],
5     size = (150, 150),
6     nx = 100,
7     ny = 100,
8     linewidth = 1,
9 )
10 X = range(xrange...; length = nx)
11 Y = range(yrange...; length = ny)
12 p = contourf(
13     X,
14     Y,
15     (x, y) -> abs(R(x + y * im)) < 1 ? -1 : 3;
16     levels = -1:1.0:2,
17     colormap = :RdYlBu_3,
18     legend = :none,
19     xrange,
20     yrange,
21     linewidth = 0,
22     size,
23     gridalpha = 1,
24     gridlinewidth = 1,
25     gridstyle = :solid,
26     aspect_ratio = 1.0,
27 )
28 hline!(p, [0]; color = :black)
29 vline!(p, [0]; color = :black)
30 p
31 end

```

Stability of the theta methods:




Explicit Euler

$$R(z) = 1 + z \xrightarrow{z \rightarrow \infty} \infty$$

Conditionally A-stable

not L-stable




Implicit Midpoint

$$R(z) = \frac{2+z}{2-z} \xrightarrow{z \rightarrow \infty} -1$$

A-stable

not L-stable




Implicit Euler

$$R(z) = \frac{1}{1-z} \xrightarrow{z \rightarrow \infty} 0$$


A-stable

L-stable


Explicit Runge-Kutta methods are conditionally A stable:



RK4




DP4




DP5

Implicit Runge-Kutta methods are A-stable, some of them are L-stable like RadauIIA(5):




```
1 stabregion(z -> R_rk(z, TableauRadauIIA(3)))
```



## Approximation order

Approximation order can generally be analysed using Taylor expansion to estimate the consistency order. We only demonstrate the results here.

```
errortest (generic function with 3 methods)
```



- Implicit Euler and explicit Euler exhibit first order of approximation
- Implicit midpoint has second approximation order
- Runge-Kutta methods can have higher order

These methods are one-step methods, which just involve two subsequent time steps. Multistep methods (e.g. BDF - backward differencing formula) achieve higher order by taking into account more of the older time steps, however they have a problem to keep their order at the start.

## An ODE system

We demonstrate the Lotka-Volterra system describing predator-prey dynamics:

- species  $x$ : prey which growth rate  $a$  being eaten with rate  $by$  by the predators
- species  $y$ : predators growing with rate  $dx$  by eating prey and dying with rate  $c$

lotkavolterra! (generic function with 1 method)

```

1 function lotkavolterra!(du, u, p, t)
2     (a, b, c, d) = p
3     x, y = u[1], u[2]
4     du[1] = a * x - b * x * y
5     return du[2] = -c * y + d * x * y
6 end

```



The function  $V$  is constant along trajectories – it is easy to show that  $\frac{dV}{dt} = 0$ . Therefore in the phase-space the trajectories are isolines of  $V(x, y)$ , and the solution must be periodic.

```

1 function V(x::Number, y::Number, param)
2     a, b, c, d = param
3     x = max(1.0e-15, x)
4     y = max(1.0e-15, y)
5     return d * x - c * log(x) + b * y - a * log(y)
6 end;

```

```
param = (a = 2, b = 6, c = 1, d = 1)
```

```
1 param = (a = 2, b = 6, c = 1, d = 1)
```


```

1 ulv = if rklv
2     explicit_rk(
3         lotkavolterra!, param; u0 = [1.0, 0.1], dt = τlv,
4         tspan = (0, 40)
5     )
6 else
7     θmethod(
8         lotkavolterra!,
9         param;
10        u0 = [1.0, 0.1],
11        dt = τlv,
12        tspan = (0, 40),
13        θ = θlv,
14        linear = linlv,
15    )
16 end;

```

linear=  RK4=

θ=  0.5 τ=  0.01



- The implicit midpoint rule catches periodicity - surprisingly also for rather coarse time steps. The reason is that it preserves more of the structure of the system. This is the simplest *geometrical integrator* - a class of methods which exactly preserves certain invariants of the solution
- The Runge-Kutta methods does a quite good job in catching periodicity

- Implicit and explicit Euler are unable to catch the dynamics in the right way.
- The linear implicit approach works well

## Stiff problems


Problems with a high ratio between the real parts of the largest and the smallest eigenvalues of  $\frac{\partial f}{\partial u}$  (spectral condition number  $\kappa$  for spd problems) are called *stiff*.

This is e.g. the case with different time scales which may be due to fast and slow reactions, and for systems arising from the discretization of PDEs, where the condition number increases with  $O(h^{-2})$ , where  $h$  is the space discretization parameter.

Stability then depends on the stability constraint for the smallest timescale.

Unconditionally A-stable and L-stable methods are to be preferred if we want to guarantee stability for stiff problems.

Let us consider a chemical reaction ("[Robertson reaction](#)" example from matlab)



fstiff! (generic function with 1 method)

```
1 function fstiff!(du, u, p, t)
2     k1, k2, k3 = p
3     du[1] = -k1 * u[1] + k3 * u[2] * u[3]
4     du[2] = k1 * u[1] - k3 * u[2] * u[3] - k2 * u[2]^2
5     return du[3] = k2 * u[2]^2
6 end
```

For this problem,  $d$  can be seen as "stiffness" parameter. The larger  $d$  the shorter the time scale of the second equation.

probertson = (k1 = 10, k2 = 1.0, k3 = 100)

```
1 probertson = (k1 = 10, k2 = k2, k3 = 100)
```


```

1 ustiff = θmethod(
2   fstiff!,
3   probertson;
4   u0 = [1.0, 0.0, 0.0],
5   dt = τx2,
6   θ = θx2,
7   linear = lin2,
8   tspan = (1.0e-6, 1.0e3)
9 );

```

$\theta =$   1.0  $\tau =$   0.027825594022071246 linear=

Stiffness  $k_2 =$   1.0



```

1 plot(
2   ustiff;
3   label = ["u[1]" "u[2]" "u[3]"],
4   size = (600, 300),
5   xlabel="t",
6   linewidth = 2,
7   framestyle = :box,
8   xscale = :log10,
9   ylimits=( -0.5, 2)
10 )

```

Due to the fact that the implicit midpoint rule is not L-stable, we can have problems with large stiffness.

## DAE problems

Differential-algebraic equation (DAE) system:

$$0 = f(\dot{u}, u, p, t)$$

Here, we will focus on mass matrix DAE systems (MMDAE):

$$M\dot{u} = f(u, p, t)$$

M is a diagonal matrix, some entries may be zero. The corresponding equations in the system turn from differential equations to algebraic equations.

We can see a DAE as the "large stiffness limit" case of an ODE system, so once again we want to use A-stable and L-stable methods.

Here, we re-define the Robertson example using the fact that the sum of all three reactants must be 1. This can be seen by adding up the three equations above resulting in  $du[1]+du[2]+du[3]=0$ . Thus  $u[1]+u[2]+u[3]$  must be constant and equal to the sum of the initial values.



fdae! (generic function with 1 method)

```
1 function fdae!(du, u, p, t)
2     k1, k2, k3 = p
3     du[1] = -k1 * u[1] + k3 * u[2] * u[3]
4     du[2] = k1 * u[1] - k3 * u[2] * u[3] - k2 * u[2]^2
5     du[3] = u[1] + u[2] + u[3] - 1
6 end
```


M\_DAE = 3x3 Diagonal{Float64, Vector{Float64}}:

```
1.0  .  .
.  1.0  .
.  .  0.0
```

```
1 M_DAE = Diagonal([1.0, 1, 0])
```

$\theta =$   1.0  $\tau =$   0.06135907273413173 linear=

Stiffness  $k_2 =$   885.8667904100827



```

1 plot(
2     udae;
3     xlabel="t",
4     label = ["u[1]" "u[2]" "u[3]"],
5     size = (600, 300),
6     linewidth = 2,
7     framestyle = :box,
8     xscale = :log10,
9     ylimits=(-0.5,1.5)
10 )

```

As for stiff problems, we want implicit methods to solve DAE problems.

## Multistep methods

Multistep methods attempt to take into account information from timesteps prior to  $t_n$ .

### BDF Methods

For a given vector  $X = x_1 \dots x_n$ , and function values  $Y = \{y_i = f(x_i)\}_{i=1 \dots n}$  a Lagrange polynomial  $L[X, Y]$  interpolates the function  $f$  in the interval  $[x_1, x_n]$ . It is given by

$$L[X, Y](\xi) = \sum_{j=1}^n l_j[X](\xi) y_j$$

with

$$l_j[X](\xi) = \prod_{i=1 \dots n, i \neq j} \frac{\xi - x_i}{x_j - x_i}$$

From the very definition we have  $l_j[X](x_i) = \delta_{ij}$ . The interesting point is that we can hope that  $L'[X](x_n)$  is a good approximation of  $f'(x_n)$ . This idea gives another way to approximate

the derivative on the left hand side of the ODE, taking into account results from older timesteps.

In particular, for the case of three interpolation points  $X = \{a, b, c\}$  we get

$$\begin{aligned} l_a[X](\xi) &= \frac{(\xi - b)(\xi - c)}{(a - b)(a - c)} & l'_a[X](\xi) &= \frac{2\xi - b - c}{(a - b)(a - c)} & l'_a[X](c) &= \frac{c - b}{(a - b)(a - c)} \\ l_b[X](\xi) &= \frac{(\xi - a)(\xi - c)}{(b - a)(b - c)} & l'_b[X](\xi) &= \frac{2\xi - a - c}{(b - a)(b - c)} & l'_b[X](c) &= \frac{c - a}{(b - a)(b - c)} \\ l_c[X](\xi) &= \frac{(\xi - a)(\xi - b)}{(c - a)(c - b)} & l'_c[X](\xi) &= \frac{2\xi - a - b}{(c - a)(c - b)} & l'_c[X](c) &= \frac{2c - a - b}{(c - a)(c - b)} \end{aligned}$$

Let  $a = t^{n-1}$ ,  $b = t^n$ ,  $c = t^{n+1}$

Setting  $Y = \{y_a, y_b, y_c\}$  gives

$$L'[X, Y](c) = \frac{1}{2\tau} (y_a - 4y_b + 3y_c)$$

For  $y_c = u^{n+1}$ ,  $y_b = u^n$ ,  $y_a = u^{n-1}$  gives rise to the BDF2 (Backward Differencing Formula) scheme:

$$\frac{1}{2} (3u^{n+1} - 4u^n + u^{n-1}) = \tau f(u^{n+1}, p, t^{n+1})$$

BDF schemes are implicit. Unlike the higher order variants, BDF1 and BDF2 are A-stable, and thus useful for stiff problems. BDF2 has consistency order 2.

It is easy to see that this procedure can be adapted to varying step sizes. Further refined variants ensure that under stepsize variations the coefficient before  $u_{n+1}$  stays unchanged (Fixed leading coefficient BDF in SUNDIALS; FBDF in Julia). More information is available via Scholarpedia.

A higher order correction of the BDF methods leads to the QNDF method (quasi constant timestep numerical differentiation formula) which is equivalent to the `ode15s` method implemented in the matlab ODE suite. QNDF is the preferred default solver for stiff systems in Julia

## Adams methods

A similar interpolation approach can be applied to the right hand side of the ODE system, giving rise e.g. to the explicit Adams-Bashforth methods.

## Rosenbrock methods

Multistage linear implicit methods (Rosenbrock methods). The simple methods of this class are

the linear implicit methods which replace the nonlinear solution by just one Newton step. Higher order variants combine a fixed number of intermediate linear steps.

## Differential equation solvers from `sciml.ai`

---

The SciML github org provides vast amount of differential equation solvers. These can be used by using `DifferentialEquations` providing all methods implemented for all sorts of differential equations (ODE, stochastic DE, delay DE)

With using `OrdinaryDiffEq` one can obtain all ODE specific methods. For faster loading it is possible now to load more specific packages.

For an overview see the [OrdinaryDiffEq documentation](#).

```
1 begin
2     using SciMLBase: ODEProblem, ODEFunction, solve
3     using OrdinaryDiffEqBDF: FBDF, QNDF2
4     using OrdinaryDiffEqRosenbrock: Rosenbrock23
5     using OrdinaryDiffEqLowOrderRK: RK4
6     using OrdinaryDiffEqSDIRK: ImplicitEuler, ImplicitMidpoint
7     using OrdinaryDiffEqFIRK: RadauIIA3
8     using DataStructures: OrderedDict
9 end
```

So far, we did our own implementations of some methods discussed. These implementations miss a number of points:

- Time step adaptivity. A general approach to time step adaptivity considers the comparison between two solutions of different consistency order to obtain an error estimate.

Julia provides a state of the art toolbox for the solution of systems of differential equations: [DifferentialEquations.jl](#). The overview on implemented methods is given in the [documentation](#):

HOME MODELING SOLVERS ANALYSIS

MACHINE LEARNING DEVELOPER TOOLS GitHub

ODE Solvers

Search...

# ODE Solvers

```
solve(prob::ODEProblem,alg;kwargs)
```

Solves the ODE defined by `prob` using the algorithm `alg`. If no algorithm is given, a default algorithm will be chosen.

## Recommended Methods

It is suggested that you try choosing an algorithm using the `alg_hints` keyword argument. However, in some cases you may want something specific, or you may just be curious. This guide is to help you choose the right algorithm.


## Unknown Stiffness Problems

When the stiffness of the problem is unknown, it is recommended you use a stiffness detection and auto-switching algorithm. These methods are multi-paradigm and allow for efficient solution of both stiff and non-stiff problems. The cost for auto-switching is very minimal, but the choices are restrained. They are a good go-to method when applicable.

We choose a subset of the methods implemented there and return to the example problems we discussed:

```
diffeqmethods = OrderedCollections.OrderedDict{String, UnionAll}({
    "Implicit Euler" => ImplicitEuler
    "Implicit Midpoint" => ImplicitMidpoint
    "RadauIIA3 (Implicit RK)" => RadauIIA3
    "RK4 (Explicit RK)" => RK4
    "Rosenbrock23 (Rosenbrock)" => Rosenbrock23
    "QNDF2" => QNDF2
    "FBDF" => FBDF
})
```





## Lotka-Volterra revisited

analyze (generic function with 1 method)

```
param2 = (a = 2, b = 6, c = 1, d = 1)
```

```
1 param2 = (a = 2, b = 6, c = 1, d = 1)
```


```
1 ulv_deq = let
2   prob = ODEProblem(lotkavolterra!, [1.0, 1.0], (0, 40), param2)
3   method = diffeqmethods[lvmethod]
4   if method == ImplicitMidpoint
5     dt = reltol
6     @info "implicit midpoint: dt=$(dt)"
7     solve(prob, method(); dt)
8   else
9     @info "reltol=$(reltol)"
10    solve(prob, method(); reltol)
11  end
12 end;
```

```
reltol=0.007054802310718643
```

method:  reltol|dt:

(timesteps = 300, smallest = 6.0503e-6, largest = 0.268243)

```
1 analyze(ulv_deq)
```



```

1 let
2   ulv = ulv_deq
3   Vulv = DiffEqArray([V(ulv[:, i]..., param2) for i in 1:length(ulv)], ulv.t)
4   puv = plot(
5     ulv[1, :],
6     ulv[2, :];
7     linewidth = 2,
8     framestyle = :box,
9     size = (200, 200),
10    label = nothing,
11    xlabel = "prey",
12    ylabel = "predators"
13  )
14  puvt = plot(
15    ulv;
16    linewidth = 2,
17    framestyle = :box,
18    gridlinewidth = 2,
19    size = (400, 200),
20    xlabel = "t",
21    label = ["prey" "predators"]
22  )
23  plot!(puvt, Vulv; linewidth = 2, color = :green, label = "V")
24  plot(puv, puvt; layout = grid(1, 2; widths = (0.3, 0.7)), size = (650, 200))
25 end

```

## Stiffness + DAE revisited

```

1 ustiff_deq = let
2   prob = ODEProblem(
3     fstiff!, [1.0, 0, 0], (1.0e-2, 10),
4     (k1 = 10, k2 = st, k3 = 100)
5   )
6   solve(prob, diffeqmethods[stmethod](); reltol = streltol)
7 end;

```

(timesteps = 61, smallest = 1.3142e-5, largest = 1.78961)

```

1 udae_deq = let
2     fdae = ODEFunction(fdae!; mass_matrix = Diagonal([1, 1, 0]))
3     prob = ODEProblem(
4         fdae, [1.0, 0, 0], (1.0e-2, 100),
5         (k1 = 10, k2 = st, k3 = 100)
6     )
7     solve(prob, diffeqmethods[stmethod](); reltol = streltol)
8 end;


```

(timesteps = 66, smallest = 1.0e-6, largest = 22.0016)

Method: Rosenbrock23 (Rosenbrock) ▾

Stiffness  $k_2 =$   10000.0

reltol:  0.0035111917342151313



## Conclusions/outlook

- There is a whole world of ODE and DAE solvers. Julia's SciML solvers cover a significant part of this world
- ODE systems created from PDE discretizations are *large* and they are *stiff*, i.e. they contain strongly differing timescales
- Stability of explicit methods is limited by that of the fastest timescale, they are not feasible for stiff problems
- Explicit methods do not work for DAEs by construction
- A- and L- stable methods are required for stiff problems

The general advise when solving a particular problem is to use the Julia's SciML solvers and to

find out which methods work best

---

## Helper cells

# Table of Contents

## ODE and DAE Solvers

- Explicit Euler method

  - Test problem

- $\theta$ - and Runge-Kutta methods

  - $\theta$  method: implementation

  - Runge-Kutta methods

- Stability

  - A-Stability

  - L-Stability

- Approximation order

- An ODE system

- Stiff problems

- DAE problems

- Multistep methods

  - BDF Methods

  - Adams methods

  - Rosenbrock methods

- Differential equation solvers from `sciml.ai`

  - Lotka-Volterra revisited

  - Stiffness + DAE revisited

- Conclusions/outlook

@statement\_str (macro with 1 method)

